
Purpose

The purpose of this document is to describe the function and operation of the Service Processor Simulator (hereafter referred to as SPSIM).  This software simulates a service processor like you might find in a machine supporting dynamic partitioning.  This service processor controls the insertion and ejection of registered devices from the machine.  In this context, a registered device is one that has been hooked into the SPSIM infrastructure.

SPSIM simulates the insertion/ejection of hardware by a service processor through these three components:

· A kernel mode driver that provides an operation region space that registered devices use to answer the _STA method.  The registered device objects also update this ‘STA space’ when the device is ejected.

· A user-mode application that allows the user to control the _STA behavior of registered devices and notify those devices as well.  The application also provides higher-level insert and eject commands.

· Sample ASL code that can be compiled into existing ACPI bioses or added dynamicly via ACPILOAD or loaded by the ACPI via the Load command.

This code needs to be modified to fit the application and is described in the Registering an ACPI device object with SPSIM section.

SPSIM also serves as a demo of writing an operation region space driver.

Kernel Mode Driver (SPSIM.SYS)

SPSIM.SYS is a function driver that will be loaded in on ACPI enumerated device objects that specify their _HID value as “SPSIMUL”.  The driver registers an operation region space that can be modified by the SPCTL application through a device interface and associated IOCTLs that the driver supports.

The ioctls supported by SPSIM.SYS are declared in the SPSIMIOCT.H file.  These IOCTLs are:

	Ioctl
	Function

	IOCTL_SPSIM_GET_MANAGED_DEVICES
	Enumerate the devices managed (registered with) SPSIM.

	IOCTL_SPSIM_ACCESS_STA
	Read/Write the STA value associated with a device.

	IOCTL_SPSIM_NOTIFY_DEVICE    
	Send an ACPI notify code to the device

	IOCTL_SPSIM_GET_DEVICE_NAME
	Get a text name for the device


SPSIM assumes the presence of the ASL infrastructure described later in this document.  The ASL defines which devices SPSIM controls, how to notify those devices, and how to describe them in the user-mode application.  The kernel mode driver provides the path through which the user-mode application can enumerate the ‘registered’ devices, modify their _STA values and generate ACPI notifications.

Note: The driver implements an additional operation region space for simulating PNP memory modules.  This code is not essential to SPSIM but is provided as an additional working example of software-implemented operation region spaces.

SPCTL

This is a user-mode command-line application that uses IOCTLs to control the operation of the simulated service processor (SPSIM.SYS).

When invoked with no command-line arguments, SPCTL dumps the list of devices controlled by the simulated service processor and the current values being provided by the driver via the operation region space for their _STA methods.

Ex:


Dev 0 : _STA is 0x0 Name \_SB_.PMM0


Dev 1 : _STA is 0xf Name \_SB_.PMM1


Dev 2 : _STA is 0x0 Name \_SB_.PMM2


Dev 3 : _STA is 0xf Name \_SB_.PMM3


Dev 4 : _STA is 0xf Name \_SB_.PMM4


Dev 5 : _STA is 0x0 Name \_SB_.PMM5


Dev 6 : _STA is 0x0 Name \_SB_.PMM6


Dev 7 : _STA is 0x0 Name \_SB_.PMM7

The first column lists the ‘devnum’ associated with each device.  This is useful in addressing that device to other SPCTL commands.  The hex number in the middle is the value currently being provided in the operation region space for that device’s _STA value.  The idea is that the ACPI device object for the registered device is modified to reference the operation region to retrieve the value it returns for it’s _STA method.  However, there is no requirement that the ACPI device object return this value directly i.e it may make it’s ‘device presence’ dependent on other factors as well, etc.

The support SPCTL commands are:

· notify devnum code

· setsta devnum value

· insert [devnum | * ]

· eject [devnum | * ]

· <no arguments> prints device status above.

Where ‘code’ is a ACPI notification code and ‘value’ is a valid result for the _STA method.  Using these primitives you can simulate successful insertion/ejection as well as surprise removal and other variations.

Any other input will result in output describing the command line usage of SPCTL.

ASL Infrastructure

SPSIM includes sample ASL code that shows SPSIM in use.  The following is a walkthrough of that ASL code indicating the function of each piece.

SPSIMUL device object

This first ASL code fragment declares a device object of HID SPSIMUL which will result in ACPI enumerating an ACPI\SPSIMUL device on which SPSIM.SYS will be loaded (if you point the Add New Hardware wizard at the appropriate INF).

        Device(XXSP) {

            Name(_HID, "SPSIMUL")

            Method(_STA) {

                return (0xF)

            }

This next fragment declares the operation region that is used to communicate _STA information between user-mode, the driver, and the registered device objects.  For each device object registered with SPSIM, there should be 8 bit area declared in the operation region.  This is the area in which the device object will store and retrieve _STA information provided/maintained by SPSIM.

            OperationRegion(DSTA, 0x99, 0, 0x10)

            Field(DSTA, ByteAcc, NoLock, Preserve) {

                SMM0, 8,

                SMM1, 8,

                SMM2, 8,

                SMM3, 8,

                SMM4, 8,

                SMM5, 8,

                SMM6, 8,

                SMM7, 8

            }

This next fragment declares a variable to be referenced to see if the SPSIM operation space is online.  Registered objects should check this value before reading/writing the STA operation region.  The _REG method will update this value when SPSIM is started or removed.

            Name(AVAL, 0)

            Method(_REG, 2) {

                Store(Arg1, AVAL)

            }

This fragment defines the SNAM package which contains a list of strings naming each device.  These names are associated on a 1:1 basis with the STA values found in the operation region field and are used only in the user-mode user-interface.

            Name(SNAM,

                Package() {

                    \_SB.PMM0,

                    \_SB.PMM1,

                    \_SB.PMM2,

                    \_SB.PMM3,

                    \_SB.PMM4,

                    \_SB.PMM5,

                    \_SB.PMM6,

                    \_SB.PMM7

                }

            )

This fragment defines NOFD method that contains a notify for each device listed in SNAM and the operation region field.  This method is used to pass on notify requests from user-mode.  The first argument is the devnum and the second argument is the notify value to be applied.

            Method(NOFD, 2) {

                if (LEqual(Arg0, 0)) {

                    Notify(\_SB.PMM0, Arg1)

                }

                if (LEqual(Arg0, 1)) {

                    Notify(\_SB.PMM1, Arg1)

                }

                if (LEqual(Arg0, 2)) {

                    Notify(\_SB.PMM2, Arg1)

                }

                if (LEqual(Arg0, 3)) {

                    Notify(\_SB.PMM3, Arg1)

                }

                if (LEqual(Arg0, 4)) {

                    Notify(\_SB.PMM4, Arg1)

                }

                if (LEqual(Arg0, 5)) {

                    Notify(\_SB.PMM5, Arg1)

                }

                if (LEqual(Arg0, 6)) {

                    Notify(\_SB.PMM6, Arg1)

                }

                if (LEqual(Arg0, 7)) {

                    Notify(\_SB.PMM7, Arg1)

                }

            }       
A ‘registered’ device object

This device object is registered with SPSIM via it’s presence in the operation region field, SNAM and its use of references to the SPSIM operation region space.

        Device(PMM0) {

Note that the _STA method has been altered to first check to see if the SPSIM STA operation region space is available before trusting any data retrieved from it.

            Method(_STA) {

                If (\_SB.XXSP.AVAL) {

                   Store(\_SB.XXSP.SMM0, Local0)

                   return (Local0)

                }

                return (0) // not here unless SPSIM is up.

            }

Similarly an _EJx method has been defined.  This method would do everything to make the device go away and not return until it had.  In this case, it notes in the operation region space that future _STA methods should return 0 i.e device not present.

            Method(_EJ0, 1) {

                Store(0, \_SB.XXSP.SMM0)

            }

        }

Registering an ACPI device object with SPSIM

To make an ACPI device object controllable from SPSIM, you make the following changes in the supplied ASL sample:

1. Add a field entry to the STA operation region space from which this device object will retrieve it’s _STA value.

2. Add a string to the SNAM package at the same index as the field in the _STA operation region space.  This name will appear in the SPCTL UI only.

3. Modify the NOFD method to include a notification statement for the device object.  Arg0 will be the 0th-based index of the device in the SNAM package and Arg1 will contain the notify code.  Ex.:

          If (Lequal(Arg0, 3)) {

              Notify(\_SB.PMM3, Arg1)

          }

4. Modify the ACPI device object’s _STA implementation to reference the field entry in the STA operation region space.  It is not required to directly return the value from the operation region, i.e. it might need to be conditional on other parameters, etc.  Don’t forget to check the AVAL variable of the SPSIMUL object to make sure the operation region is valid before believing any data from it.

5. Modify the ACPI device object’s _EJx or _RMV implementation to make appropriate changes to the device’s _STA value via the operation region space as appropriate.

Installation

The SPSIMUL object needs to be added to your ACPI namespace either by modifying the ACPI bios or by loading additional tables via ACPILOAD.







